PME 3380 – Modelagem de Sistemas Dinâmicos

LISTA B

Tiago Vieira de Campos Krause 9836238

Equações diferenciais:

Exemplo 1:

O exemplo 1 envolve a aplicação de um algoritmo de integração numérica usando o método de Euler, cujo algorítmo segue a seguinte função:

$$h(t + dt) = h(t) + dt \cdot \dot{h}(t)$$

O algoritmo é aplicado para a equação diferencial ordinária linear $\dot{y}(t) = [1 - y(t)]/2$, com instante inicial $t = 0$, instante final $t_f = 10$ e condição inicial $y(0) = 0$. Sendo a solução analítica exata $y_e(t) = 1 - e^{-t/2}$.

Os seguintes gráficos são obtidos com o plot da solução analítica e exata:





Exemplo 2:

O exemplo 2 envolve a aplicação de um algoritmo de integração numérica usando o método de Runge Kutta, cujo algorítmo segue a função $h(t + dt)$:

$$k_1 = \dot{h}(h(t), t)$$

$$k_2 = \dot{h}(h(t) + k_1 \cdot dt/2, t + dt/2)$$

$$k_3 = \dot{h}(h(t) + k_2 \cdot dt/2, t + dt/2)$$

$$k_4 = \dot{h}(h(t) + k_3 \cdot dt, t + dt)$$

$$h(t + dt) = h(t) + dt \cdot (k_1 + 2k_2 + 2k_3 + k_4)/6$$

O algoritmo é aplicado para a mesma equação diferencial do exemplo 1 e tem as mesmas condições e solução analítica.

Os seguintes gráficos são obtidos com o plot da solução analítica e exata:

Graphic window number 0



Graphic window number 1



Exercício 1:

Implemente um programa no Scilab que resolva numericamente a equação diferencial que modela o sistema abaixo, tanto pelo método de Euler como Runge Kutta.



Reservatório com água
Parâmetros:
S = 10 m^2 - área da seção transversal (constante)
$R = 2\times10^8$ $Pa/(m^3/s)^2$ - parâmetro que relaciona vazão com queda de pressão (perda de carga)
ρ = 1000 kg/m^3 - massa específica da água
G = 10 m/s^2 - aceleração da gravidade na superfície da terra

Variáveis:
Q_e = 0,010247 m^3/s - vazão de entrada
h: nível do reservatório [m]
V: volume de água no reservatório [m^3]
P: pressão relativa à atmosférica, no fundo do reservatório [Pa]
Q_s: vazão de saída [m^3/s]

Admite-se que a água seja incompressível.

Pela equação da continuidade:

$$\frac{dV}{dt} = Q_e - Q_s$$

Vamos admitir que a perda de carga na saída é modelada pela expressão:

$$P = RQ_s^2 \quad \Rightarrow \quad Q_s = \sqrt{\frac{P}{R}}$$

Por outro lado, a pressão no fundo do reservatório é:

$$P = \rho g h$$

Considere uma entrada Q_e constante.

Volume de água no reservatório:

$$V = Sh \quad \Rightarrow \quad \dot{V} = S\dot{h}$$

Substituindo:

$$S\dot{h} = Q_e - \sqrt{\frac{\rho g h}{R}}$$

Resultando na seguinte equação difererencial ordinária não linear (modelo de 1 reservatório):

$$\dot{h} = \left(-\sqrt{\frac{\rho g h}{R}} + Q_e\right)\frac{1}{S}$$

Analisando o sistema de reservatório de água apresentado e aplicando à solução numérica a equação diferencial ordinária não linear $\dot{h} = \left(-\sqrt{\frac{\rho g h}{R}} + Q_e\right)\frac{1}{S}$ fornecida no exercício obtém-se o seguinte código:

```
// Conjunto de comandos para solucao numerica de equacao diferencial
// Apagando dados anteriores:
clear
// Definindo a equacao diferencial:
function [ydot]=funcao(y)
ydot=(-(p*g*y/R)^0.5+Qe)/S;
endfunction
// Instante inicial:
t(1)=0;
// Instante final:
tf=60000;
// Condicao inicial:
//Método de Euler:
ye(1)=2;
//Método de Runge Kutta:
yrk(1)=2;
// Passo de integracao (experimente alterar o passo):
h=2;
// Definição dos parâmetros:
S = 10 //m2 - área da seção transversal
R = 2*10^8 //Pa/(m3/s)2 - parâmetro que relaciona vazão com queda de pressão (perda de carga)
p = 1000 //kg/m3 - massa específica da água
g = 10 //m/s2 - aceleração da gravidade na superfície da terra
Qe = 0.010247 //m3/s - vazão de entrada
//y: nível do reservatório [m]
//V: volume de água no reservatório [m3]
//P: pressão relativa à atmosférica, no fundo do reservatório [Pa]
//Qs: vazão de saída [m3/s]
```

```
// Calculo de numero de passos):
n=round(tf/h);
// Integracao numerica usando o metodo de Euler:
    // Comando for:
    for i=1:n
    // Vetor de tempo:
    t(i+1)=t(i)+h;
    // Solucao numerica pelo método de euler:
    ye(i+1)=ye(i)+h* funcao(ye(i));
    // Solução numerica pelo método de Runge Kutta
    k1=(-(p*g*yrk(i)/R)^0.5+Qe)/S;
    k2=(-(p*g*(yrk(i)+k1*h/2)/R)^0.5+Qe)/S;
    k3=(-(p*g*(yrk(i)+k2*h/2)/R)^0.5+Qe)/S;
    k4=(-(p*g*(yrk(i)+k3*h)/R)^0.5+Qe)/S;
    yrk(i+1)=yrk(i)+(h*(k1+2*k2+2*k3+k4)/6);
    // Termino do comando for:
    end
// Plotando solucao numerica y versus vetor de tempo t:
plot2d([t,t],[ye, yrk],[1,2]);
// Colocando uma legenda na parte inferior direito da figura (parametro 4):
legends(["Método de Euler", "Método de Runge Kutta"],[1,2],4)
// Colocando um titulo na figura e nomeando os eixos:
xtitle("Nível do reservatório [m] vs tempo","Tempo t","Nível do reservatório")
```

Considerou-se a condição inicial do nível do reservatório sendo igual a 2m para obter uma convergência mais rápida. Variando o passo h é possível observar como cada método se comporta em cada caso. Para tf=40000s e h=100, 1000 e 2000s, respectivamente, obtêm-se os seguintes gráficos:


Nível do reservatório [m] vs tempo
Nível do reservatório
Tempo t
Método de Euler
Método de Runge Kutta
2
2.01
2.02
2.03
2.04
2.05
2.06
2.07
2.08
2.09
2.1
2.11
2.12
0
5 000
10 000
15 000
20 000
25 000
30 000
35 000
40 000


Nível do reservatório [m] vs tempo
Nível do reservatório
Tempo t
Método de Euler
Método de Runge Kutta
2
2.01
2.02
2.03
2.04
2.05
2.06
2.07
2.08
2.09
2.1
2.11
2.12
0
5 000
10 000
15 000
20 000
25 000
30 000
35 000
40 000



Plotando as curvas obtidas com passo 100s e 2000s em um mesmo gráfico a seguir e focando na região de convergência pode-se comparar os resultados e observar que ambos os métodos convergem para o valor de 2.1m mas a alteração do passo possui certa influência sobre o método de Euler, sendo que o método de Runge Kutta representa melhor o processo mesmo com um passo maior.



Exercício 2:

Desenvolva um programa em Scilab que resolva numericamente o sistema de equações diferencias que modela o sistema com dois reservatórios, usando tanto Euler como Runge Kutta. Dica: raciocine com vetores.



Modelo do sistema de 2 reservatórios (considere a entrada constante e perdas de carga não lineares como no caso do ex. de 1 tanque).

$$\begin{cases} \dot{h}_1 = \left[Q_e - \sqrt{\dfrac{\rho g}{R_a}(h_1 - h_2)} \right] \dfrac{1}{S_1} \\ \dot{h}_2 = \left[\sqrt{\dfrac{\rho g}{R_a}(h_1 - h_2)} - \sqrt{\dfrac{\rho g}{R_s} h_2} \right] \dfrac{1}{S_2} \end{cases}$$

Para este caso gerou-se o seguinte código para aplicar ambos os métodos de solução numérica em ambas as equações diferenciais indicadas no exercício:

```
// Conjunto de comandos para solucao numerica de equacao diferencial
// Apagando dados anteriores:
clear
// Definindo a equacao diferencial:
function [ydot]=funcao1(y1,y2)
ydot=(-(p*g*(y1-y2)/R1)^0.5+Qe)/S1;
endfunction

function [ydot]=funcao2(y1,y2)
ydot=((p*g*(y1-y2)/R1)^0.5-(p*g*y2/R2)^0.5)/S2;
endfunction

// Instante inicial:
t(1)=0;
// Instante final:
tf=60000;
// Condicao inicial:
//Método de Euler:
ye1(1)=0;
ye2(1)=0;
//Método de Runge Kutta:
yrk1(1)=0;
yrk2(1)=0;
// Passo de integracao (experimente alterar o passo):
h=100;
// Definição dos parâmetros:
S1 = 10 //m2 - área da seção transversal do reservatório 1
S2 = 10 //m2 - área da seção transversal do reservatório 2
R1 = 2*10^8 //Pa/(m3/s)2 - parâmetro que relaciona vazão com queda de pressão (perda de carga) na saída do reservatório 1
R2 = 2*10^8 //Pa/(m3/s)2 - parâmetro que relaciona vazão com queda de pressão (perda de carga) na saída do reservatório 2
p = 1000 //kg/m3 - massa específica da água
g = 10 //m/s2 - aceleração da gravidade na superfície da terra
Qe = 0.010247 //m3/s - vazão de entrada
//y: nível do reservatório [m]
//V: volume de água no reservatório [m3]
//P: pressão relativa à atmosférica, no fundo do reservatório [Pa]
//Qs: vazão de saída [m3/s]
```

```
// Calculo de numero de passos):
n=round(tf/h);
// Integracao numerica usando o metodo de Euler:
    // Comando for:
    for i=1:n
    // Vetor de tempo:
    t(i+1)=t(i)+h;
    // Solucao numerica pelo método de euler:
    ye1(i+1)=ye1(i)+h* funcao1(ye1(i),ye2(i));
    ye2(i+1)=ye2(i)+h* funcao2(ye1(i),ye2(i));
    // Solução numerica pelo método de Runge Kutta
    //Para o reservatório 1 (k11, k21, k31, k41) e para o reservatório 2 (k12, k22, k32, k42)
    k11=funcao1(yrk1(i),yrk2(i))
    k12=funcao2(yrk1(i),yrk2(i))
    k21=funcao1(yrk1(i)+k11*h/2,yrk2(i)+k12*h/2)
    k22=funcao2(yrk1(i)+k11*h/2,yrk2(i)+k12*h/2)
    k31=funcao1(yrk1(i)+k21*h/2,yrk2(i)+k22*h/2)
    k32=funcao2(yrk1(i)+k21*h/2,yrk2(i)+k22*h/2)
    k41=funcao1(yrk1(i)+k31*h,yrk2(i)+k32*h)
    k42=funcao2(yrk1(i)+k31*h,yrk2(i)+k32*h)
    yrk1(i+1)=yrk1(i)+(h*(k11+2*k21+2*k31+k41)/6);
    yrk2(i+1)=yrk2(i)+(h*(k12+2*k22+2*k32+k42)/6);
    // Termino do comando for:
    end
// Plotando solucao numerica y versus vetor de tempo t:
plot2d([t,t,t,t],[ye1, ye2, yrk1, yrk2],[1,2,3,4]);
// Colocando uma legenda na parte inferior direito da figura (parametro 4):
legends(["Reservatório 1 - Euler", "Reservatório 2 - Euler","Reservatório 1 - RK", "Reservatór
io 2 - RK"],[1,2,3,4],4)
// Colocando um titulo na figura e nomeando os eixos:
xtitle("Nível do reservatório [m] vs tempo","Tempo t","Nível do reservatório")
```

Rodando o algorítmo variando a condição inicial para 0m em ambos os reservatórios e para 3m em ambos, e variando o passo para 100 e 1000s obtém-se os seguintes gráficos:

Passo h=100s, condição inicial 0m:



Passo h=1000s, condição inicial 0m:



Passo h=100s, condição inicial 3m:



Passo h=1000s, condição inicial 3m:



Pode-se observar que os métodos convergem para o mesmo nível de reservatório mesmo variando as condições iniciais e também que para um passo maior o método de Euler representa o processo com menos precisão comparado ao método de Runge Kutta.